

5 イジング模型の相転移 (齋藤 幸夫)

5.1 統計力学、相転移、臨界現象

水は熱すれば水蒸気となり、冷やせば氷となる。水と水蒸気では密度が 1000 倍以上異なるし、水と氷では流動性が全然違っている。これらは多数の水分子の集まりが示す特別の性質を持った状態、「相」の例であり、水蒸気は水の「気相」、水は「液相」、氷は「固相」に付けられた呼び名である。(これに対し、お湯というのは温かい「液相」であり、別の特別な相を意味してはいない。) 水蒸気の温度を下げていくと、普通、水蒸気 → 水 → 氷という変化が起きるが、これは「相転移」の一種である。

現在の情報化社会を支えるシリコンなどの半導体には、純度の良い結晶が求められている。それには半導体の融液からの結晶成長という相転移を上手に制御することが大切である。また、集積回路を高密度に書き込むためには、結晶表面を原子レベルで制御することが必要であり、それには気相からの結晶成長という相転移が用いられている。金属や酸化物セラミックスは低温で電気抵抗ゼロの超伝導相になるが、これに大電流を流して強磁場を発生させると、磁気浮上の高速リニアモーターカーに用いられる。逆に超伝導状態での磁束の量子化により非常に弱い磁場を測れるので、医学用の MRC T スキャンに用いられている。フェライト等の磁性体は常温で強磁性体であり、磁気を帯びているので、磁石となる。この磁石の微粉末は磁気テープや自動改札用の切符の裏など、情報記録媒体として用いられている。自発的に電気分極を持つ強誘電体には、機械的圧力をかけると電気を発生する圧電効果を持つ物があり、超音波発生や各種リモコンの赤外線発生源として使われている。このように物質が相転移すると、色々な新しい性質を持った相が見つかることがある。

この相転移は多数の原子や分子の間の相互作用に依って引き起こされる。熱力学で習った通り、その第二法則に依れば、あらゆる物質は自由エネルギー G の低い状態に向かう。ここで原子間のエネルギーを H 、原子がバラバラで自由な配置をとりたいというエントロピーを S とすると、温度 T での自由エネルギーは $G = H - TS$ と与えられる。従って、温度 T が高いところでは右辺第二項が重要となって、相互作用 H が無いときの状態、例えば「気相」が実現する。一方、温度が下がると第一項が重要となって、エネルギーの低い状態、例えば「固相」が実現される。統計力学では原子間の微視的な相互作用 \mathcal{H} が与えられれば、分配関数 (または状態和) $Z = \text{Tr} e^{-\mathcal{H}/k_B T} = e^{-G/k_B T}$ の計算をする事で自由エネルギー G が求められる。

さて、普通、液体は温度を上げると気体になる。ところがある圧力以上ではどんなに温めても気体になりえず、ただ密度が減るだけという現象が起きる。この気化の起きない限界の圧力、温度の点を臨界点という。液体-気体の共存線に沿って温度、圧力を上げていくと、液体と気体の密度の差が小さくなっていて、臨界点では密度は同じになってしまう。この様に、臨界点に近づくとき色々な異常が

みられ、臨界異常と呼ばれる。この臨界現象を調べるには統計力学の分配関数が計算されれば良いのだが、特別のモデルを除いては厳密解が求められない。普通は、近似的解析法しかない。けれど計算機を用いると、大きさが有限の多粒子系の熱平衡状態を定量的に詳しく調べられる。

5.2 モンテカルロ法

コンピューターによる相転移研究の例として、イジング強磁性体を取りあげる。一辺の長さが L の正方格子を考え、格子定数を 1 とすると、全体で $N = L^2$ 個の格子点がある。その各格子点 i に、外部磁場 h と同じまたは逆の向きを向ける、小さな磁石 σ_i がある。この σ をスピンと呼び、 h と同じ向きの時を $\sigma = +1$ とし、逆向きの時を $\sigma = -1$ とする。そして h と σ が同じ向きならばエネルギーは下がり、逆向きならばエネルギーが上がるものとする。そのほか、隣合うスピンの向きが同じならばエネルギーは $-J$ 下がり、逆向きならば $+J$ 上がるとする。これらをまとめて書けば

$$\mathcal{H} = -J \sum_{\langle ij \rangle} \sigma_i \sigma_j - h \sum_{i=1}^N \sigma_i$$

となる。ここで $\langle ij \rangle$ は格子点上隣合う全ての格子点の組についての和を取ることを表す。

さて、系の微視的状态 S はスピンの配列 $(\sigma_1, \sigma_2, \dots, \sigma_N)$ できまる。それは

$$\begin{aligned} S_1 &= (+ + + \cdots +) \\ S_2 &= (- + + \cdots +) \\ S_3 &= (+ - + \cdots +) \\ &\vdots \\ S_{2^N} &= (- - - \cdots -) \end{aligned}$$

のように、全部で 2^N 個ある。状態 S のエネルギーを $\mathcal{H}(S)$ とすると、熱平衡ではこの状態が実現する確率は $P_{\text{eq}}(S) = Z^{-1} e^{-\mathcal{H}(S)/k_B T}$ で与えられる。ここで

$$Z = \sum_{\sigma_1=\pm 1} \cdots \sum_{\sigma_N=\pm 1} e^{-\mathcal{H}/k_B T} = \sum_{n=1}^{2^N} e^{-\mathcal{H}(S_n)/k_B T}$$

は状態和である。物理量 $A(S)$ の熱平衡値は $\langle A \rangle \equiv \sum_S A(S) P_{\text{eq}}(S)$ で与えられる。そこで、全ての状態 S を作りだし、熱平衡の確率 $P_{\text{eq}}(S)$ を掛けて平均値を計算すれば良い。しかしサイト数 N が増えると可能な状態数 2^N はネズミ算的に膨大な物になり、計算機の容量を越えてしまう。そこで、平衡分布の重みに従って、状態を発生させてやる方法が考え出され、これがモンテカルロ法である。

モンテカルロ法では次々に状態を確率的に変化させていく。その m ステップ目に状態 S が出現している確率を $P(S; m)$ とする。次の $m+1$ ステップ目に状態 S

が出現する確率 $P(S; m+1)$ は

$$P(S; m+1) = P(S; m) - \sum_{S'} W(S \rightarrow S') P(S; m) + \sum_{S'} W(S' \rightarrow S) P(S'; m)$$

のように書ける。ここで $W(S \rightarrow S')$ は状態 S から状態 S' へ移る遷移確率で、右辺第二項は状態 S から S' へ逃げ出して行くための確率 $P(S)$ の減少を表し、第三項は状態 S' から S へやってきたための確率の増加を表す。熱平衡では確率は変化しなくなるので第二項と第三項はバランスしているはずである。そのためには遷移確率が

$$\frac{W(S \rightarrow S')}{W(S' \rightarrow S)} = \frac{P_{\text{eq}}(S')}{P_{\text{eq}}(S)} = \exp \left[-\frac{\mathcal{H}(S') - \mathcal{H}(S)}{k_B T} \right]$$

という詳細釣り合いの条件を満たしていれば良い。例えば Metropolis の方法では、状態変化 $S \rightarrow S'$ に伴うエネルギー変化 $\Delta E = \mathcal{H}(S') - \mathcal{H}(S)$ を求め、

$$\begin{aligned} \Delta E \leq 0 \quad \text{なら} \quad W(S \rightarrow S') &= 1, \\ \Delta E > 0 \quad \text{なら} \quad W(S \rightarrow S') &= \exp(-\Delta E/k_B T) \end{aligned}$$

とする。

さて、 m ステップ目の状態 $S^{(m)}$ から $m+1$ ステップ目の状態 $S^{(m+1)}$ を生成するシミュレーションの概略は以下の通りである。

- (1) m ステップ目の状態 $S^{(m)} = S$ が与えられているとする。
- (2) 以下の (i)–(iii) の手続きを N 回繰り返し、得られた N 回後の状態を $S^{(m+1)}$ とする。
 - (i) 何処か一つの格子点を選んで、そのスピンを反転した状態を S' とする。
 - (ii) $W(S \rightarrow S')$ を計算する。
 - (iii) 0 と 1 の間に一様分布する乱数 r を計算し、

$$\begin{aligned} r \leq W(S \rightarrow S') \quad \text{ならば状態を } S' \text{ に変える。} \\ r > W(S \rightarrow S') \quad \text{ならば状態は } S \text{ のままにする。} \end{aligned}$$

このようにして生成した状態列 $(S^{(0)}, S^{(1)}, \dots, S^{(m)}, \dots, S^{(M_f)})$ は $M_f \rightarrow \infty$ で熱平衡の分布に従って出現していることが保証されている。

さて、物理量 A の熱平均値は

$$\langle A \rangle = (M_f - M_i)^{-1} \sum_{m=M_i+1}^{M_f} A(S^{(m)})$$

で与えられる。ここで M_i は、初期状態の効果をなくすために、平均値の計算の中には取り入れないモンテカルロステップ数である。例えば、平均の磁化 $\langle M \rangle$ とエネルギー $\langle E \rangle$ は

$$\langle M \rangle = (M_f - M_i)^{-1} \sum_{m=M_i+1}^{M_f} \left(\sum_{i=1}^N \sigma_i^{(m)} \right)$$

$$\langle E \rangle = (M_f - M_i)^{-1} \sum_{m=M_i+1}^{M_f} \mathcal{H}(S^{(m)})$$

で計算される。又、帯磁率 χ と比熱 C はそれぞれ磁化とエネルギーの揺らぎから

$$\chi = \frac{\partial \langle M \rangle}{\partial h} = \frac{\langle M^2 \rangle - \langle M \rangle^2}{k_B T} = \frac{1}{k_B T} \left[\frac{\sum_{m=M_i+1}^{M_f} (\sum_{i=1}^N \sigma_i^{(m)})^2}{M_f - M_i} - \langle M \rangle^2 \right]$$

$$C = \frac{\partial \langle E \rangle}{\partial T} = \frac{(\langle E^2 \rangle - \langle E \rangle^2)}{k_B T^2} = \frac{1}{N T^2} \left[\frac{\sum_{m=M_i+1}^{M_f} \mathcal{H}(S^{(m)})^2}{M_f - M_i} - \langle E \rangle^2 \right]$$

で与えられる。

シミュレーションで実現した事象は統計的にばらつきのある現象から取りだしたものであるため揺らぎを持っており、得られた平均値にも統計的な誤差がつきまわっている。この統計誤差を評価するには、作り出された状態列を、完全に相関がなくなっていると思われる L 個ずつのグループ K 個 ($M_f - M_i = LK$) に分ける。各グループの平均値が全体の平均値からどれくらい揺らぐかの偏差を求めて、統計誤差を評価する。例えば磁化 $\langle M \rangle$ の統計誤差は

$$\sigma_M^2 = \frac{1}{K-1} \left[K^{-1} \sum_{k=1}^K (L^{-1} \sum_{m=M_i+(k-1)L+1}^{M_i+kL} \sum_{i=1}^N \sigma_i^{(m)})^2 - \langle M \rangle^2 \right]$$

により与えられる。

ところで、二次元のイジング模型に対しては $h = 0$ の時には厳密解が分かっており、正方格子では相転移の起こる転移温度 T_c やその他の物理量が以下のように求められている。

$$\begin{aligned} \frac{k_B T_c}{J} &= \frac{2}{\ln(\sqrt{2} + 1)} \simeq 2.269 \\ -\frac{F}{N k_B T} &= \ln(2 \cosh 2K) + \frac{1}{2\pi} \int_0^\pi \ln \left[\frac{1}{2} (1 + \sqrt{1 - k^2 \cos^2 \omega}) \right] d\omega \\ -\frac{E}{N J} &= \coth 2K \left[1 + \frac{2}{\pi} k'' K(k) \right] \\ \frac{C}{N k_B} &= K^2 \frac{2}{\pi} (\coth 2K)^2 \left[2K(k) - 2E(k) - (1 - k'') \left(\frac{\pi}{2} + k'' K(k) \right) \right] \end{aligned}$$

ここで温度の逆数は

$$K = \frac{J}{k_B T}$$

と表され、パラメター k と k'' は

$$k = \frac{2 \sinh 2K}{\cosh^2 2K} \quad k'' = 2 \tanh^2 2K - 1$$

を表し、 $K(k)$ と $E(k)$ は以下のような第一種及び第二種の楕円積分を表す。

$$K(k) = \int_0^{\pi/2} \frac{d\omega}{\sqrt{1 - k^2 \sin^2 \omega}} \quad E(k) = \int_0^{\pi/2} \sqrt{1 - k^2 \sin^2 \omega} d\omega$$

特に、相転移点付近では物理量の温度依存性に色々な異常が現れるという、臨界現象がみられる。例えば、磁化 m は T_c 以上ではゼロであるが、 T_c 以下では有限の値を持つ。この磁化を自発磁化と呼び、この状態を強磁性状態と呼ぶ。自発磁化の温度依存性は T_c 近傍では

$$m \sim (T_c - T)^\beta \quad \beta = 1/8$$

となる。また比熱は T_c で発散し、

$$C \sim \ln |T - T_c| \sim |T - T_c|^{-\alpha} \quad \alpha = 0$$

である。同様に、帯磁率も T_c で発散して、その近くでは

$$\chi \sim |T - T_c|^{-\gamma} \quad \gamma = 7/4$$

である。これら、物理量の臨界点近くでの異常を示す指数 α, β, γ などを臨界指数という。物理量 A の臨界異常を表す指数 μ を求めるには、 A の温度依存性を両対数表示して、以下の直線に合わせることで

$$\ln A \sim \mu \ln |T - T_c| + C_0$$

その傾きから μ が推定される。但し上式が成立するのは無限に大きな系 ($N \rightarrow \infty$) の場合であり、 N の有限な系では A のサイズ N に対する依存性からも μ の推定が出来る (有限サイズ相似則)。しかしこれについては本書の程度を越えるので略する。

5.3 プログラム例

2次元正方格子上のイジング模型の磁場の無い場合のモンテカルロ・シミュレーションのプログラム例が `/home/kyozai/jy/comp-phys/sec5/{ising.f90,rnd.f90}` である。ising.f90 がモンテカルロ・シミュレーションのプログラム本体で、rnd.f90 は ising.f90 の中で使っている乱数生成のサブルーチンである。

このプログラムでは、 $S \rightarrow S'$ の遷移として、ランダムに一つのスピン σ_i を選びそれを反転する場合を考えている。即ち、 S から σ_i を反転して得られるスピン配置が S' である。この遷移に対する遷移確率 $W(S \rightarrow S')$ を $W_i(S)$ と書くことにする。 $S = (\sigma_1, \sigma_2, \dots, \sigma_N)$ とすると

$$\mathcal{H}(S) = -J\sigma_i \sum_{j \text{ は } i \text{ と隣接}} \sigma_j - J \sum_{\langle j,k \rangle, j \neq i, k \neq i} \sigma_j \sigma_k$$

と書けるので、


$$\Delta E = \mathcal{H}(S') - \mathcal{H}(S) = 2J\sigma_i \sum_{j \text{ は } i \text{ と隣接}} \sigma_j = 2J\sigma_i e_i$$

となる。正方格子の場合、 $\sigma_i e_i$ は $-4, -2, 0, 2, 4$ の値をとる。メトロポリス法の場合、

$$W_i(S) = \begin{cases} 1 & \cdots \sigma_i e_i \leq 0 \\ \exp\left(-\frac{2J}{k_B T} \sigma_i e_i\right) & \cdots \sigma_i e_i > 0 \end{cases}$$

となる。

プログラム全体は次のような構成になっている。



では、配列 `ispin(0:lmax-1,0:lmax-1)`, `ip(0:lmax-1)`, `im(0:lmax-1)`, `trprob(-nnno:nnno)` を宣言している。

配列 `ispin(0:lmax-1,0:lmax-1)` の要素 `ispin(ix, iy)` には、位置 (ix, iy) でのスピンの値 ± 1 が格納される。ここで、正方格子は $lsize \times lsize$ の大きさで、 ix, iy は $0, 1, 2, \dots, lsize - 1$ の値をとる。 $lmax$ は $lsize$ の最大値で `parameter` 文で定義されている。端の影響を取り除くために、周期的境界条件を用いる。即ち、 $(0, iy)$ と $(lsize-1, iy)$ は隣接している。同様に、 $(ix, 0)$ と $(ix, lsize-1)$

は隣接している。この周期的境界条件のもとで、配列 $\text{ip}(0:\text{lmax}-1)$ の要素 $\text{ip}(i)$ には座標 i を 1 増やした隣の座標が、配列 $\text{im}(0:\text{lmax}-1)$ の要素 $\text{im}(i)$ には座標 i を 1 減らした隣の座標が格納される。

配列 $\text{trprob}(-\text{nnno}:\text{nnno})$ の要素 $\text{trprob}(ie)$ には


$$\sigma_i e_i \equiv \sigma_i \sum_{j \text{ は } i \text{ と隣接}} \sigma_j = ie$$

の時の σ_i の反転する遷移確率 $W_i(S)$ が格納される。 nnno は最隣接格子点の数で `parameter` 文で 4 と設定されている。

● パラメーター入力


では、以下の変数の入力、検査、出力を行なっている。

| | |
|---------------------|---|
| <code>lsize</code> | 正方格子の一辺の長さ |
| <code>temp</code> | 温度 (無次元量 $k_B T/J$ の値) |
| <code>inist</code> | 初期スピン配置を決めるパラメーター $\begin{cases} -1: & \text{全ての } \sigma_i = -1 \\ 0: & \text{確率 } 1/2 \text{ で } \sigma_i = \pm 1 \\ 1: & \text{全ての } \sigma_i = 1 \end{cases}$ |
| <code>imcs</code> | 初期に捨てるモンテカルロ・ステップ数 |
| <code>nmcs</code> | 1 ブロック当りのモンテカルロ・ステップ数 |
| <code>nblock</code> | ブロック数 |
| <code>iseed</code> | 乱数を生成するサブルーチンを初期化するためのパラメーター、乱数の種 |



● 初期設定


では、



のように初期化を行なっている。

- 1 MC ステップ

では、前節の (1), (2) のようにモンテカルロ法を実行している。



- ブロック和初期化

では、ブロック内の各ステップで測定した量の総和を計算するための変数を 0 にしている。

| | |
|-------|--|
| amag | imag = $\sum_i \sigma_i$ の総和用 |
| amag2 | imag の 2 乗の総和用 |
| aene | iene = $\sum_{\langle i,j \rangle} \sigma_i \sigma_j$ の総和用 |
| aene2 | iene の 2 乗の総和用 |

これらの変数は、ブロックごとに物理量 (磁化、帯磁率、エネルギー、比熱) を計算する際に必要になる。

- 測定

では、上記の imag, iene を計算している。

- ブロック和加算

では、imag, imag の 2 乗, iene, iene の 2 乗を amag, amag2, aene, aene2 に加えている。

• ブロック平均

では、`imag`, `imag` の 2 乗, `iene`, `iene` の 2 乗の 1 ブロック内での平均を計算し出力している。この出力は後に統計処理に使うことができる。さらに、磁化 M 、帯磁率 χ 、エネルギー E 、比熱 C に対して、次の量 (スピン 1 個当りの無次元量) を計算している。

$$\begin{aligned} \text{avm} &= \frac{M}{N} = \frac{1}{N} \langle \sum_i \sigma_i \rangle_{\text{MC}} \\ \text{avx} &= \frac{\chi k_B T}{N} = \frac{1}{N} \left\{ \left\langle \left(\sum_i \sigma_i \right)^2 \right\rangle_{\text{MC}} - \left\langle \sum_i \sigma_i \right\rangle_{\text{MC}}^2 \right\} \\ \text{ave} &= \frac{E}{NJ} = -\frac{1}{N} \langle \sum_{\langle i,j \rangle} \sigma_i \sigma_j \rangle_{\text{MC}} \\ \text{avc} &= \frac{C}{N k_B} = \frac{1}{N} \left(\frac{J}{k_B T} \right)^2 \left\{ \left\langle \left(\sum_{\langle i,j \rangle} \sigma_i \sigma_j \right)^2 \right\rangle_{\text{MC}} - \left\langle \sum_{\langle i,j \rangle} \sigma_i \sigma_j \right\rangle_{\text{MC}}^2 \right\} \end{aligned}$$

ここで、 $\langle \cdots \rangle_{\text{MC}}$ はブロックごとのモンテカルロ法による平均を表している。

ising.f90 のリスト

```

=====
! Monte Carlo simulation of 2-D Ising model (no field)
!   originally written by Y. Saito
!   modified for the f90 form by J.Yamauchi 2011-10-20
=====
program ising
!=====
!  sengen
!=====
implicit none
integer,parameter:: lmax=64, nnno=4
integer:: lsize,inist,imcs,nmcs,nblock,iseed &
,ispin(0:lmax-1,0:lmax-1),ip(0:lmax-1),im(0:lmax-1)
real(kind=8):: temp &
,traprob(-nnno:nnno)
!
integer:: no,lsize1,i,ix,iy,nm,n,nb,ie,iene,imag
real(kind=8):: de,amag,amag2,aene,aene2,ave,avc,dene,avm,avx,dmag,rnd
!=====
!  parameter nyuryoku
!=====
!  nyuryoku
!=====
write(7,*) 'linear size?'
read(*,*) lsize
!
write(7,*) 'temperature?'
read(*,*) temp
!
write(7,*) 'initial configuration?'
write(7,*) ' -1: all down'
write(7,*) '  0: random'
write(7,*) '  1: all up'
read(*,*) inist
!
write(7,*) 'number of initial Monte Carlo steps?'
read(*,*) imcs
write(7,*) 'number of Monte Carlo steps per block?'
read(*,*) nmcs
write(7,*) 'number of blocks?'
read(*,*) nblock
!
write(7,*) 'seed of random numbers (odd positive integer)?'
read(*,*) iseed
!=====
!  kensa
!=====
if((lsize.gt.lmax).or.(lsize.lt.1)) then
  write(*,*) 'error in lsize:',lsize
  stop
endif
if((inist.lt.-1).or.(inist.gt.1)) then
  write(7,*) 'error in inist:',inist
  stop
endif
if((iseed.le.0).or.(mod(iseed,2).eq.0)) then
  write(7,*) 'error in iseed:',iseed
  stop
endif
if(imcs.lt.0) then
  write(7,*) 'error in imcs:',imcs
  stop
endif
if(nmcs.lt.0) then
  write(7,*) 'error in nmcs:',nmcs
  stop
endif
if(nblock.lt.0) then
  write(7,*) 'error in nblock:',nblock
  stop
endif
!=====
!  shutsuryoku
!=====
write(*, '(a10,i12)') ' lsize= ',lsize
write(*, '(a10,e24.16)') ' temp= ',temp
write(*, '(a10,i12)') ' inist= ',inist
write(*, '(a10,i12)') ' imcs= ',imcs
write(*, '(a10,i12)') ' nmcs= ',nmcs
write(*, '(a10,i12)') ' nblock= ',nblock
write(*, '(a10,i12)') ' iseed= ',iseed

```

```

=====
! shokisettei
=====
! ransu shokika
=====
call rndini(iseed)
=====
! tonari no hyo shokika
=====
no=lsiz*lsiz
lsiz=lsiz-1
do i=0,lsiz-1
  ip(i)=i+1
  im(i)=i-1
enddo
ip(lsiz)=0
im(0)=lsiz-1
=====
! spin haichi shokika
=====
do ix=0,lsiz-1
  do iy=0,lsiz-1
    if(inist.ge.1) then
      ispin(ix,iy)=1
    else if(inist.le.-1) then
      ispin(ix,iy)=-1
    else
      call rndu(rnd)
      if(rnd.ge.0.5d0) then
        ispin(ix,iy)=1
      else
        ispin(ix,iy)=-1
      endif
    endif
  enddo
enddo
=====
! sen'i-kakuritsu shokika
=====
do n=-nnno,nnno
  de=2.0d0*dble(n)/temp
  if(de.gt.0.0d0) then
    trprob(n)=dexp(-de)
  else
    trprob(n)=1.0d0
  endif
enddo
=====
! shoki loop
=====
do nm=1,imcs
  =====
  ! 1 Monte Carlo step
  =====
  do n=1,no
    call rndu(rnd)
    ix=lsiz*rnd
    call rndu(rnd)
    iy=lsiz*rnd
    ie=ispin(ix,iy)*(ispin(ip(ix), iy)+ispin(im(ix), iy) &
      +ispin( ix ,ip(iy))+ispin( ix ,im(iy)))
    call rndu(rnd)
    if(rnd.lt.trprob(ie)) then
      ispin(ix,iy)=-ispin(ix,iy)
    endif
  enddo
enddo

```

```

!=====
! block loop
!=====
do nb=1,nblock
!=====
!   block-wa shokika
!=====
amag =0.0d0
amag2=0.0d0
aene =0.0d0
aene2=0.0d0
!=====
!   Monte Carlo step loop
!=====
do nm=1,nmcs
!=====
!       1 Monte Carlo step
!=====
do n=1,no
call rndu(rnd)
ix=lsiz*rnd
call rndu(rnd)
iy=lsiz*rnd
ie=ispin(ix,iy)*(ispin(ip(ix), iy)+ispin(im(ix), iy) &
+ispin( ix ,ip(iy))+ispin( ix ,im(iy)))
call rndu(rnd)
if(rnd.lt.trprob(ie)) then
ispin(ix,iy)=-ispin(ix,iy)
endif
enddo
!=====
!       sokutei
!=====
iene=0
imag=0
do ix=0,lsiz1
do iy=0,lsiz1
imag=imag+ispin(ix,iy)
iene=iene+ispin(ix,iy)*(ispin(ip(ix),iy)+ispin(ix,ip(iy)))
enddo
enddo
!=====
!       block-wa kasan
!=====
aene =aene +dble(iene)
aene2=aene2+dble(iene)*dble(iene)
amag =amag +dble(imag)
amag2=amag2+dble(imag)*dble(imag)
enddo
!=====
!       block heikin
!=====
aene =aene /dble(nmcs)
aene2=aene2/dble(nmcs)
amag =amag /dble(nmcs)
amag2=amag2/dble(nmcs)
write(*,'(a10,i12)') '    block= ',nb
write(*,'(a10,e24.16)') '    aene= ',aene
write(*,'(a10,e24.16)') '    aene2= ',aene2
write(*,'(a10,e24.16)') '    amag= ',amag
write(*,'(a10,e24.16)') '    amag2= ',amag2
dene =aene2-aene*aene
ave  =-aene/dble(no)
avc  =(dene/dble(no))/(temp*temp)
dmag =amag2-amag*amag
avm  =amag/dble(no)
avx  =dmag/dble(no)
write(*,'(a10,e24.16)') '    e= ',ave
write(*,'(a10,e24.16)') '    c= ',avc
write(*,'(a10,e24.16)') '    m= ',avm
write(*,'(a10,e24.16)') '    x= ',avx
enddo
!
stop
end program ising

```

5.4 補足 (高野 宏)

5.4.1 カノニカル分布

体積 V , 粒子数 N の系が絶対温度 T の熱浴と接している場合、エネルギー E_n をもつ系の微視的状态 n の実現確率は

$$P_n = \frac{\exp(-\beta E_n)}{Z}$$

で与えられる。ただし、 $\beta = \frac{1}{k_B T}$ で

$$Z = \sum_n \exp(-\beta E_n)$$

は分配関数と呼ばれる。この分布をカノニカル分布という。

系の自由エネルギー $F(T, V, N)$ は、分配関数 Z によって

$$F = -\frac{1}{\beta} \ln Z$$

で与えられる。系の内部エネルギーは、この分布に関するエネルギーの平均値として

$$\langle E \rangle = \frac{\sum_n E_n \exp(-\beta E_n)}{Z} = \frac{-\frac{\partial}{\partial \beta} Z}{Z} = -\frac{\partial}{\partial \beta} \ln Z = \frac{\partial}{\partial (1/T)} \frac{F}{T}$$

で与えられる。定積比熱 C は

$$\begin{aligned} C &= \frac{\partial \langle E \rangle}{\partial T} \\ &= -k_B \beta^2 \frac{\partial \langle E \rangle}{\partial \beta} \\ &= -k_B \beta^2 \frac{\partial}{\partial \beta} \frac{\sum_n E_n \exp(-\beta E_n)}{Z} \\ &= k_B \beta^2 \left[\frac{\sum_n E_n^2 \exp(-\beta E_n)}{Z} - \left(\frac{\sum_n E_n \exp(-\beta E_n)}{Z} \right)^2 \right] \\ &= k_B \beta^2 (\langle E^2 \rangle - \langle E \rangle^2) \end{aligned}$$

で与えられる。

系のハミルトニアン \mathcal{H} に変数 x とそれに共役な物理量が $\mathcal{H} = \mathcal{H}_0 + xX$ の形で含まれており、 $E_n(x) = E_n(x=0) + X_n x$ となる場合 (\mathcal{H}_0 と X が同時固有状態をもつ場合) を考える。 X の状態 n における値は $X_n = \frac{\partial E_n}{\partial x}$ で与えられ、カノニカル分布での X の平均値は、

$$\langle X \rangle = \frac{\sum_n X_n \exp(-\beta E_n)}{Z} = \frac{\sum_n \left(\frac{\partial E_n}{\partial x} \right) \exp(-\beta E_n)}{Z} = \frac{1}{\beta} \frac{\partial}{\partial x} \ln Z = \frac{\partial F}{\partial x}$$

で与えられる。 X の x に対する感受率 χ は

$$\begin{aligned}\chi &= \frac{\partial \langle X \rangle}{\partial x} \\ &= \frac{\partial}{\partial x} \frac{\sum_n X_n \exp(-\beta E_n)}{Z} \\ &= \beta \left[\frac{\sum_n X_n^2 \exp(-\beta E_n)}{Z} - \left(\frac{\sum_n X_n \exp(-\beta E_n)}{Z} \right)^2 \right] \\ &= \beta (\langle X^2 \rangle - \langle X \rangle^2)\end{aligned}$$

で与えられる。ただし、 $\frac{\partial X_n}{\partial x} = 0$ を用いた。

5.4.2 モンテカルロ法

モンテカルロ法は、確率的にサンプリングすることにより積分を数値的に実行する方法である。例えば、積分 $I = \int_a^b f(x) dx$ は、積分区間 $[a, b]$ に一様に分布する確率変数 r を用いて

$$I = (b - a) \int_a^b f(r) \frac{1}{b - a} dr = (b - a) \langle f(r) \rangle$$

のように、 r に関する平均で表すことができる。この平均を N 個の r のサンプル r_1, r_2, \dots, r_N を発生させ、

$$I = (b - a) \langle f(r) \rangle \simeq (b - a) \frac{1}{N} \sum_{i=1}^N f(r_i)$$

のように評価することができる。このように、積分領域を一様にサンプリングする方法を単純サンプリングという。この評価法は被積分関数が積分領域内にほぼ一様に分布している場合に有効と考えられる。

系のエネルギーが、系の状態を記述する変数 \mathbf{x} で $E = \mathcal{H}(\mathbf{x})$ で与えられていたとする。カノニカル分布における物理量 $A(\mathbf{x})$ の平均

$$\langle A \rangle = \frac{\sum_{\mathbf{x}} A(\mathbf{x}) \exp(-\beta \mathcal{H}(\mathbf{x}))}{\sum_{\mathbf{x}} \exp(-\beta \mathcal{H}(\mathbf{x}))}$$

を計算するのに、単純サンプリングを用いて、状態空間中からランダムに M サンプル

$\mathbf{x}_1, \mathbf{x}_2, \dots, \mathbf{x}_M$ をとり、

$$\langle A \rangle = \frac{\sum_{i=1}^M A(\mathbf{x}_i) \exp(-\beta \mathcal{H}(\mathbf{x}_i))}{\sum_{i=1}^M \exp(-\beta \mathcal{H}(\mathbf{x}_i))}$$

のように計算することも原理的には可能である。しかし、系の自由度が大きい場合、

$\exp(-\beta\mathcal{H}(\mathbf{x}))$ は状態空間の中で非常に激しく変化する関数で、単純サンプリングでは精度が悪くなる。そこで、カノニカル分布での状態の出現確率 $P_{\text{eq}}(\mathbf{x}) = Z^{-1} \exp(-\beta\mathcal{H}(\mathbf{x}))$ に比例した確率で状態のサンプルを $\mathbf{x}_1, \mathbf{x}_2, \dots, \mathbf{x}_M$ のように M 個とり、

$$\langle A \rangle = \sum_{\mathbf{x}} P_{\text{eq}}(\mathbf{x}) A(\mathbf{x}) = \frac{1}{M} \sum_{i=1}^M A(\mathbf{x}_i)$$

のように $\langle A \rangle$ を計算する方法が考えられた。このようなサンプリングの方法を重み付きサンプリングという。

このような重み付きのサンプルを作るのに、ステップ n での状態 $\mathbf{x}(n)$ からステップ $n+1$ での状態 $\mathbf{x}(n+1)$ を確率的に生成するというマルコフ鎖の方法が用いられる。状態 \mathbf{x} が1ステップ後に状態 \mathbf{x}' に遷移する確率を $W(\mathbf{x} \rightarrow \mathbf{x}')$ とする。 n ステップで系が状態 \mathbf{x} にある確率を $P_n(\mathbf{x})$ と書くと、

$$P_{n+1}(\mathbf{x}) = P_n(\mathbf{x}) + \sum_{\mathbf{x}'} [-W(\mathbf{x} \rightarrow \mathbf{x}')P_n(\mathbf{x}) + W(\mathbf{x}' \rightarrow \mathbf{x})P_n(\mathbf{x}')]]$$

と書ける。この $P_n(\mathbf{x})$ が $n \rightarrow \infty$ でカノニカル分布 $P_{\text{eq}}(\mathbf{x}) = Z^{-1} \exp(-\beta\mathcal{H}(\mathbf{x}))$ になるためには、次の2条件を満たせば十分である。

(1) エルゴード性:

任意の状態 \mathbf{x} から任意の状態 \mathbf{x}' へ有限回の遷移で到達できる。

(2) 詳細釣り合:

遷移確率が任意の状態 \mathbf{x}, \mathbf{x}' に対して

$$W(\mathbf{x} \rightarrow \mathbf{x}')P_{\text{eq}}(\mathbf{x}) = W(\mathbf{x}' \rightarrow \mathbf{x})P_{\text{eq}}(\mathbf{x}')$$

を満たす。

詳細釣り合を満たす遷移確率として、よく使われるものに次のものがある。

1. メトロポリス法

$$W(\mathbf{x} \rightarrow \mathbf{x}') = \begin{cases} 1 & \cdots \mathcal{H}(\mathbf{x}') \leq \mathcal{H}(\mathbf{x}) \\ \exp[-\beta(\mathcal{H}(\mathbf{x}') - \mathcal{H}(\mathbf{x}))] & \cdots \mathcal{H}(\mathbf{x}') > \mathcal{H}(\mathbf{x}) \end{cases}$$

2. 熱浴法

$$W(\mathbf{x} \rightarrow \mathbf{x}') \propto \frac{\exp[-\beta\mathcal{H}(\mathbf{x}')] }{\exp[-\beta\mathcal{H}(\mathbf{x}')] + \exp[-\beta\mathcal{H}(\mathbf{x})]}$$

5.4.3 擬似乱数

一様な乱数を擬似的に生成する方法として Lehmer 法 (linear congruential method) がある。それは、

$$R_{n+1} = aR_n + c \pmod{M}$$

に従って正又は 0 の整数列 R_n を生成する方法である。サンプル rnd.f90 では、 $a = 69069$, $c = 0$, $M = 2^{32}$ を乱数の初期化サブルーチン rndini で用いている。この場合、 R_n の初期値 (乱数の種) は奇数でなければならない。乱数の周期は 2^{30} となる。

Tausworthe 法 (Maximum-length linearly recurring sequence) では

$$R_n = R_{n-q} \oplus R_{n-p}$$

に従って整数列 R_n が生成される。ここで、 \oplus は整数の各ビットごとの排他的論理和で、1 ビットだけ考えると $1 \oplus 1 = 0 \oplus 0 = 0$, $1 \oplus 0 = 0 \oplus 1 = 1$ となる。HP の fortran では整数 i1 と i2 の各ビットごとの排他的論理和は ieor(i1, i2) で与えられる。サンプル rnd.f90 のサブルーチン rndu では $q = 32$, $p = 521$ の Tausworthe 法を用いている。これにより、 $0 \sim 2^{31} - 1$ の整数の乱数が生成され、 $[0, 1)$ の倍精度実数の一様乱数に変換されている。乱数の周期は $2^{521} - 1$ となる。

5.4.4 有限サイズスケーリング

一辺が L の d 次元スピン系を考える。温度を T 、磁場を h 、臨界温度を T_c とする。このとき、系のスピン当たりの自由エネルギーの臨界点近傍の異常をあたえる部分 $f(T, h, L)$ は臨界点近傍 ($T \sim T_c$, $h \sim 0$) では次のように振舞うことが有限サイズスケーリングから予想されている。

$$f(T, h, L) \simeq L^{-d} \hat{f}(L^{y_T}(T - T_c), L^{y_H}h)$$

比熱 $c \sim \frac{\partial^2}{\partial T^2} f$ は

$$c(T, h, L) \simeq L^{2y_T-d} \hat{c}(L^{y_T}(T - T_c), L^{y_H}h)$$

となり、 $L \rightarrow \infty$ で L 依存性がなくなることより、

$$c(T, h = 0) \sim (T - T_c)^{-\alpha} \sim (T - T_c)^{-2+(d/y_T)}$$

が得られる。すなわち比熱の発散の臨界指数は $\alpha = 2 - (d/y_T)$ となる。同様に磁化 $m \sim \frac{\partial}{\partial h} f$ と帯磁率 $\chi \sim \frac{\partial^2}{\partial h^2} f$ に対して

$$m(T, h, L) \simeq L^{y_H-d} \hat{m}(L^{y_T}(T - T_c), L^{y_H}h)$$

$$\chi(T, h, L) \simeq L^{2y_H-d} \hat{\chi}(L^{y_T}(T - T_c), L^{y_H}h)$$

が得られ、 $L \rightarrow \infty$ において、

$$\begin{aligned} m(T = T_c, h) &\sim h^{1/\delta} \sim h^{(d-y_H)/y_H} \\ m(T, h = 0) &\sim (T_c - T)^\beta \sim (T_c - T)^{(d-y_H)/y_T} \\ \chi(T, h = 0) &\sim |T - T_c|^{-\gamma} \sim |T - T_c|^{-(2y_H-d)/y_T} \end{aligned}$$

が得られる。これより臨界指数は $\delta = y_H/(d - y_H)$, $\beta = (d - y_H)/y_T$, $\gamma = (2y_H - d)/y_T$ となる。2次元のイジング模型では $y_T = 1$, $y_H = 15/8$ である。

有限サイズスケーリングより臨界点、臨界指数を評価することができる。例えば、帯磁率は臨界点で

$$\chi(T = T_c, h = 0, L) \simeq L^{2y_H-d}$$

のように L とともに発散するので、臨界点でいろいろな L に対して帯磁率を計算することにより y_H を評価できる。臨界温度がわからない場合、三つの異なる大きさ $L_1 < L_2 < L_3$ の系に対して帯磁率を計算し、

$$\frac{\ln(\chi(T, h = 0, L_2)/\chi(T, h = 0, L_1))}{\ln(L_2/L_1)} \quad \text{と} \quad \frac{\ln(\chi(T, h = 0, L_3)/\chi(T, h = 0, L_2))}{\ln(L_3/L_2)}$$

という二つの量を計算する。有限サイズスケーリングからは二つの量は臨界温度で一致し、その値は $2y_H - d$ となることが予想される。そこで、二つの量を温度の関数としてプロットし、その交点から臨界温度と $2y_H - d$ が評価できる。また、交点付近の曲線の傾きから y_T を評価することもできる。

5.4.5 有限系のイジング模型の厳密解

周期的境界条件を課した2次元正方格子上的イジング模型の有限系の厳密解が磁場の無い場合に知られている。その表式は原論文 (B. Kaufman: Phys. Rev. **76** (1949) 1232) に与えられている。/phys/work/comp-phys/samples/sec5/exact.f90 は、この厳密解に基づき $L \times L$ の正方格子上的磁場の無いイジング模型のエネルギーと比熱を計算するプログラムである。プログラムの指示に従い L と $k_B T/J$ を入力すると、 E/NJ と C/Nk_B を出力する。プログラムは入力終了まで次々に入力を受け付けるので、端末からパラメータを入力している場合、プログラムを終了するには CTRL-D を入力して入力終了をプログラムに知らせる。リダイレクションによってファイルからパラメータを入力している場合、ファイルのおわりが入力の終了となり、プログラムが終了する。

5.4.6 統計誤差の推定

確率変数 A の独立なサンプルが a_1, a_2, \dots, a_n のように n 個ある場合、 A の平均値と分散の推定値は

$$\langle A \rangle = \frac{1}{n} \sum_{i=1}^n a_i$$

$$\langle A^2 \rangle - \langle A \rangle^2 = \frac{1}{n-1} \left[\sum_{i=1}^n a_i^2 - \frac{1}{n} \left(\sum_{i=1}^n a_i \right)^2 \right]$$

で与えられる。 n が大きくなると n 個のサンプルの平均値は、平均が母集団の平均、分散が母集団の分散の $1/n$ のガウス分布に近付くと考えられるので、 $\langle A \rangle$ の推定値 $\frac{1}{n} \sum_{i=1}^n a_i$ には $\pm \sqrt{\frac{1}{n-1} \left[\sum_{i=1}^n a_i^2 - \left(\frac{1}{n} \sum_{i=1}^n a_i \right)^2 \right]}$ の誤差が予想される。

モンテカルロ・シミュレーションでブロック毎の平均値が独立と見なせるかどうかを調べるには、ブロックを複数個ずつまとめて新しいブロックと見なした時の誤差が変化するかどうかを調べればよい。ブロック毎の平均値が独立ならば、誤差はほとんど変化しないはずである。

5.4.7 X ウィンドウ上にスピン配置を描くプログラム

/home/kyozai/jy/comp-phys/sec5 というディレクトリの下のサンプルプログラム ising.f90 を X ウィンドウ上にスピン配置を描きながら走るように改めたものが同じディレクトリの下に isingx.f90 です。isingx.f90 はメインプログラムで、このほかに乱数発生サブルーチンの rnd.f90 と FORTRAN で書かれたメインプログラムから X ウィンドウ上に図形を描くための C で書かれたサブルーチンの xsub.c を使用します。

これらのプログラムのコンパイル、リンクの方法は 3.5.2 節の説明と同じです。これらのファイルを自分のディレクトリにコピーした後、まず xsub.c をコンパイルします。

```
% gcc -I/usr/include/X11 -c xsub.c ↵
```

とタイプすることにより、xsub.c がコンパイルされ、オブジェクト・ファイル xsub.o が作られます。次に isingx.f90, rnd.f90 をコンパイルします。

```
% gfortran -fno-underscoring -c isingx.f90 rnd.f90 ↵
```

とタイプすることにより、isingx.f90, rnd.f90 がコンパイルされ、オブジェクト・ファイル isingx.o, rnd.o が作られます。最後に、これらをリンクします。

```
% gfortran isingx.o rnd.o xsub.o -lX11 ↵
```

とタイプすることにより、isingx.o, rnd.o, xsub.o および X ウィンドウ用のサブルーチンがリンクされ実行可能ファイル a.out が作られます。isingx.f90, rnd.f90 のコンパイルと最後のリンクは

```
% gfortran -fno-underscoring isingx.f90 rnd.f90 xsub.o -lX11 ↵
```

とタイプすることにより一度に行なうこともできます。

次に、このプログラムを実行します。理工学 ITC のワークステーション・ルームのワークステーションからリモートログインして計算サーバを使っているとき、このプログラムを計算サーバで走らせて理工学 ITC のワークステーションの X ウィンドウの画面上に図形を描かせることができます。そのための準備に関しては、3.5.1 節の説明を参照してください。

5.5 課題

- 5-A. サンプル・プログラムを参考にして、ブロック毎の平均をさらに平均するようにして、 8×8 の格子で磁場の無い場合に 1 ～ 3 個の温度で内部エネルギー、比熱を計算し、有限系の厳密解と比較せよ。
- 5-B. 16×16 の格子に対して磁場の無い場合に内部エネルギー、比熱、磁化、帯磁率の温度変化を計算しグラフにプロットせよ。統計誤差も評価してグラフにエラーバーとして表示せよ。
- 5-C. $L \times L$ の格子に対して磁場の無い場合に $L = 8, 16, 32$ に対して帯磁率の温度変化を計算し有限サイズスケーリングを用いて臨界温度 T_c と臨界指数 y_H を評価せよ。また、厳密な値と比較せよ。有限系で自発磁化 $\langle M \rangle / N$ が 0 であることを考慮し、帯磁率として $\chi k_B T / N = \langle M^2 \rangle / N$ を用いよ。